
ΑΣΚΗΣΗ 4: ΑΛΗΘΕΙΑ, ΕΛΕΓΧΟΣ & ΟΛΑ ΤΑ ΥΠΟΛΟΙΠΑ
Μεταφέρετε τα παρακάτω δεδομένα στην CLIPS.
(deftemplate personal-data


(slot name)


(slot age)


(slot weight)


(slot smoker)


(multislot date-of-birth)

)

(deffacts people


(personal-data (name adam) (weight 60) (age 30) 



(smoker no) (date-of-birth 18 06 1970))


(personal-data (name brenda) (weight 120) (age 45) 



(smoker yes) (date-of-birth 18 06 1955))


(personal-data (name charles) (weight 120) (age 60)



(smoker yes)(date-of-birth 18 06 1940))

)

(deffacts data


(date 26 04 2007)

)

Προσθέστε και τρέξτε τον παρακάτω κσνόνα.
(defrule cardiac-risk


(personal-data (name ?name) (smoker yes) (weight ?weight))


(test (> ?weight 100))

=>


(assert (cardiac-risk ?name))

)

Ο κανόνας αυτός ελέγχει εάν κάποιος είναι υπέρβαρος και καπνίζει. Εάν είναι, τότε προσθέτει στην λίστα των γεγονότων μια προειδοποίηση ότι η καρδίά του βρίσκεται σε κίνδυνο. Τι γίνεται όμως στην περίπτωση που κάποιος χάσει κιλά ή σταματήσει το κάπνισμα; Το προειδοποιητικό γεγονός θα παραμείνει στην λίστα γεγονότων, γιατί δεν είναι συνδεδεμένο με τον κανόνα που το δημιούργησε.  Αυτό σημαίνει ότι η λίστα γεγονότων δεν είναι αληθής. Μπορούμε να αντιμετωπίσουμε αυτό το πρόβλημα χρησιμοποιώντας το στοιχείο (logical).
Κάντε (clear) στο σύστημά σας.
Εισάγετε πάλι την δομή των δεδομένων σας, και τα παραπάνω γεγονότα.
Προσθέστε τον κανόνα cardiac-risk ως εξής:
(defrule cardiac-risk


(logical (personal-data (name ?name) (smoker yes) (weight ?weight)))


(logical (test (> ?weight 100)))

=>


(assert (cardiac-risk ?name))

)

Όταν τρέχουμε τον κανόνα τα αποτελέσματα είναι τα ίδια με αυτά της προηγούμενης εκδοχής (προστίθενται στην λίστα γεγονότων τα γεγονότα (cardiac-risk brenda) και (cardiac-risk charles)). Τρέξτε τον κανόνα για να το δείτε. 

Η διαφορά φαίνεται εάν μεταβάλλουμε τα αρχικά δεδομένα. Βρείτε το νούμερο (τον κωδικό) του γεγονότος των προσωπικών δεδομένων της Brenda (π.χ. 2) και πληκτρολογήστε την εντολή 

CLIPS> (modify 2 (weight 80))

Παρατηρήστε ότι το γεγονός (cardiac-risk brenda) εξαφανίστηκε από την λίστα γεγονότων. Με το στοιχείο logical δημιουργήθηκε μια σύνδεση ανάμεσα στο γεγονός που δημιουργήθηκε και τα αίτια χάρη στα οποία δημιουργήθηκε. Όταν τα αίτια αυτά αλλάζουν, τα αποτελέσματά τους δεν ισχύουν πλέον και απομακρύνονται. 
Γιατί να μην χρησιμοποιούμε το (logical) παντού; Πρώτον, αυξάνει την απαιτούμενη μνήμη και τον χρόνο επεξεργασίας. Δεύτερον, δεν χρειάζεται πάντα. Συχνά τα αρχικά δεδομένα είναι ικανοποιητικά. 
ΕΠΙΛΥΣΗ ΣΕΙΡΑΣ ΠΥΡΟΔΟΤΗΣΗΣ ΚΑΝΟΝΩΝ

Όλοι οι κανόνες των οποίων οι συνθήκες ικανοποιούνται εισάγονται στην agenda.
Εάν στην agenda υπάρχουν περισσότεροι από ένας κανόνες, η σειρά πυροδότησής τους επιλέγεται με βάση 2 κριτήρια: 

· την προτεραιότητα των κανόνων, και

· τη στρατηγική επίλυσης συγκρούσεων
Διαφορετικά, αν δεν οριστεί κανένα από τα δύο αυτά κριτήρια, η σειρά πυροδότησης θα είναι τυχαία.
 ΠΡΟΤΕΡΑΙΟΤΗΤΑ ΚΑΝΟΝΩΝ

Για τον έλεγχο της σειράς πυροδότησης των κανόνων υπάρχει η ιδιότητα (salience). Κατά την δήλωση ενός κανόνα, μπορεί να δηλωθεί και μια τιμή (salience) για τον κανόνα μεταξύ -10000 και 10000. Όσο μεγαλύτερη η τιμή αυτή, τόσο μεγαλύτερη και η προτεραιότητα του κανόνα (τόσο πιο γρήγορα πυροδοτείται ο κανόνας). Εάν δεν δηλωθεί κάποια τιμή (salience), ο κανόνας παίρνει αυτόματα την προκαθορισμένη τιμή 0. 
Πληκτρολογήστε τους παρακάτω δύο κανόνες: 

(defrule poke-fun-at-smokers


(personal-data (name ?name) (smoker yes))

=>


(printout t ?name " is a fool." crlf)

)

(defrule worry-about-thin-people


(personal-data (name ?name) (weight ?weight))


(test (< ?weight 80))

=>


(printout t ?name " is looking a bit thin." crlf))

Εάν τρέξουν αυτοί οι κανόνες, θα πυροδοτηθούν και οι δύο, χωρίς να μπορούμε να ελέγξουμε την σειρά πυροδότησης. Εάν, όμως, τους τροποποιήσουμε ως εξής: 

(defrule poke-fun-at-smokers


(declare (salience 10)) 


(personal-data (name ?name) (smoker yes))

=>


(printout t ?name " is a fool." crlf))

(defrule worry-about-thin-people


(declare (salience 20)) 


(personal-data (name ?name) (weight ?weight))


(test (< ?weight 80))

=>


(printout t ?name " is looking a bit thin." crlf))

τότε ο η μεγαλύτερη τιμή salience του κανόνα worry-about-thin-people θα εξασφαλίσει ότι θα πυροδοτηθεί πρώτος. 
ΣΤΡΑΤΗΓΙΚΗ ΕΠΙΛΥΣΗΣ ΣΥΓΚΡΟΥΣΕΩΝ
Το CLIPS διαθέτει 7 στρατηγικές επίλυσης συγκρούσεων, οι βασικότερες των οποίων είναι οι εξής:

- depth

Οι νέοι (τελευταίοι) κανόνες μπαίνουν "πάνω" από τους "παλαιούς" στην agenda, και πυροδοτείται ο πιο τελευταίος πρώτος (LIFO). To πόσο «καινούριος» ή «παλιός» είναι ένας κανόνας καθορίζεται από το πότε εισήχθησαν τα γεγονότα που ικανοποιούν τον κανόνα στη λίστα γεγονότων.

- breadth
Οι νέοι (τελευταίοι) κανόνες μπαίνουν "κάτω" από τους "παλαιούς" στην agenda, και πυροδοτείται ο πιο παλιός πρώτος (FIFO).

- simplicity

Προτεραιότητα δίνεται στον πιο «απλό» κανόνα, στον κανόνα που περιλαμβάνει τον μικρότερο αριθμό συγκρίσεων στο αριστερό του μέρος. Συγκρίσεις είναι η σύγκριση με μια σταθερά ή με την τιμή μιας δεσμευμένης μεταβλητής, η κλήση μιας συνάρτησης. 
- complexity
Προτεραιότητα δίνεται στον πιο «σύνθετο» κανόνα, στον κανόνα που περιλαμβάνει τον μεγαλύτερο αριθμό συγκρίσεων.
- random
Τυχαία πυροδότηση.

ΠΑΡΑΔΕΙΓΜΑ ΕΠΙΛΥΣΗΣ ΣΥΓΚΡΟΥΣΕΩΝ
(deffacts sample-data (data 3))

(defrule start-rule


(initial-fact)


(data ?x)

 =>

  
(assert (temp2 (* 2 ?x)))

  
(assert (temp3 (* 3 ?x))))

(defrule test2


(temp2 ?x)

 =>

  
(printout t "double: " ?x crlf))

(defrule test3


(temp3 ?x)

 =>

  
(printout t "triple: " ?x crlf))

CLIPS> (set-strategy depth)

depth

CLIPS> (reset)

CLIPS> (run)

triple: 9

double: 6

CLIPS> (set-strategy breadth)

depth

CLIPS> (reset)

CLIPS> (run)

double: 6

triple: 9
ΒΡΟΧΟΙ ΚΑΙ ΕΛΕΓΧΟΣ
Εκτός από χειρισμό γεγονότων και κανόνων, το CLIPS προσφέρει ένα πλήρες σύνολο προγραμματιστικών λειτουργιών που προσφέρουν και οι κλασσικές διαδικαστικές γλώσσες προγραμματισμού. 

Πληκτρολογήστε το παρακάτω τμήμα κώδικα:
(assert (num 1)) 
(defrule numbers

(num ?n)

=>

(loop-for-count (?n 1 9) do


(printout t “the num is: ” ?n crlf)))

Και τρέξτε τον κανόνα. 
Πληκτρολογήστε τώρα τον κανόνα:

(defrule numbers2

(num ?n)

=>

(while (< ?n 10)

do

(printout t “the num is: ” ?n crlf)

(bind ?n (+ 1 ?n))))
και τρέξτε τον.
Τέλος, πληκτρολογήστε τον κανόνα:

(defrule sign
(num ?n)

=>

(if (> ?n 0)

   then (printout t “positive” crlf)

else (if (< ?n 0)

   then (printout t “negative” crlf)

else (printout t “zero” crlf))))
και τρέξτε τον δίνοντας διάφορες τιμές στη num.
ΑΣΚΗΣΗ
Φτιάξτε ένα πρόγραμμα που να υπολογίζει το κατάλληλο δώρο για έναν άνθρωπο λαμβάνοντας υπόψη την ηλικία, την μόρφωση του, τις προτιμήσεις του, και την τιμή του δώρου.  Χρησιμοποιήστε γεγονότα της μορφής
(age 12)

(agegroup old)
(agegroup child)

(agegroup middle)
(educated yes)
(price expensive)
Χρησιμοποιώντας κανόνες της παρακάτω μορφής μπορείτε να ομαδοποιήσετε ηλικίες:

(defrule child "child age rule"

   (age ?x)

      =>


(if (< ?x 15) then

   (assert (agegroup child))))
Παρόμοιους κανόνες μπορείτε να χρησιμοποιήσετε για να να καθορίσετε εάν ένα δώρο είναι ακριβό ή φτηνό με βάση την τιμή του. Παίξτε με 4-5 διαφορετικά δώρα.

